# Parametric Curves

Natacha Fontes-Merz, Kris Green, and Sarah Raynor

Susie the rebellious clownfish is swimming away from the school one day. Sushi's position as a function of time is given by

$$\mathbf{x}(t)=(2\sin(t)\cos(2t),2\sin(t)\sin(2t),2\cos(t))$$

We are going to use Maple to understand her travels.

```
> restart:with(plots):
```

## Visualizing her Path

Our first goal is to plot her path. We need to define **x**(t) for Maple:

```
> x:=t->2*sin(t)*cos(2*t);
  y:=t->2*sin(t)*sin(2*t);
  z:=t->2*cos(t);
  starttime:=0;
  endtime:=2*Pi;
```

$$x := t \to 2\sin(t)\cos(2t)$$

$$y := t \to 2\sin(t)\sin(2t)$$

$$z := t \to 2\cos(t)$$

$$starttime := 0$$

$$endtime := 2\pi \quad (1.1)$$

We are now set to plot her motion in the xy-plane:

```
> plot([x(t),y(t),t=starttime..endtime]);
```




We can also animate the curve, which will allow us to see where she starts and which way she travels:

```
> animatecurve([x(t),y(t),t=starttime..endtime],numpoints=1000,
  frames=10);
```




Click on the graph and press the play button above to see the motion.

## Exercises

**#1: Give your best description of the motion based on the xy-plot above:**

**#2: Use the commands above to plot Susie's motion in the yz-plane and the xz-plane.**

>

**#3: Now that you have looked at Susie's motion in all three coordinate planes, do your best to describe what her motion really looks like in 3d.**

Now that we have studied Susie's motion in the coordinate planes, let's try to plot her motion in 3d:

```
> spacecurve([x(t),y(t),z(t),t=starttime..endtime],numpoints=
  1000,scaling=constrained,axes=normal);
```




You can click on the plot to rotate the angle of view. You can also set the exact angle you want to look at in the menu at the upper left. Try viewing it at theta=0, phi=90. Does this view look familiar? Rotate the plot so that you are looking at it from each of the three angles we viewed above. Here is an animation of her motion:

```
> animate(spacecurve,[[x(t),y(t),z(t)],t=starttime..x,
  numpoints=1000,axes=normal,scaling=constrained],x=starttime..
  endtime,frames=10);
```

x = 0.




## Exercises

# Susie meets a friend?

At time $t = 0$, Susie spots a shark in the distance at (100, -50, 0). The shark is swimming with a constant velocity of

$$\mathbf{v}(t) = (-25/\pi, 25/2\pi, 1/2\pi).$$

If we plot both the path of the shark and the path of Susie, we can determine whether their paths cross.

## Do the paths cross?

First we parameterize the path of the shark. Recall that the parameterization of a line through the point **a** and in the direction of vector **v** is simply **z**(t) = **a** + t***v**.

```
> x2 := t -> 100 - (25/Pi)*t;
  y2 := t -> -50 + (25/(2*Pi))*t;
  z2 := t -> 0 + (1/(2*Pi))*t;
```

$$x2 := t \to 100 - \frac{25\,t}{\pi}$$

$$y2 := t \to -50 + \frac{25}{2}\,\frac{t}{\pi}$$

$$z2 := t \to \frac{1}{2}\,\frac{t}{\pi} \qquad (2.1.1)$$

Then we graph the two paths together.

```
> dist := spacecurve([x(t)-x2(t), y(t)-y2(t), z(t)-z2(t), t = starttime .. 3*endtime], numpoints = 100, scaling = constrained, axes = normal, view = [-3 .. 3, -3 .. 3, -3 .. 3],color=black):

  display({dist});
```




```
> susie := spacecurve([x(t), y(t), z(t), t = starttime .. 3*endtime], numpoints = 1000, scaling = constrained, axes = normal, view = [-3 .. 3, -3 .. 3, -3 .. 3],color=black):
  shark := spacecurve([x2(t), y2(t), z2(t), t = starttime .. 3*endtime], numpoints = 1000, scaling = constrained, axes =
```

```
normal, view = [-3 .. 3, -3 .. 3, -3 .. 3],color=red):
display({shark, susie});
```




Note that in the graph, we have limited the plotting region, in order to magnify the motion of the fish and show the two paths together. Otherwise, the shark's path covers a huge region.

### *Exercises*

**#7)** It appears from the graph that the shark does intersect Susie's path. At what point do the two paths intersect? Be sure to justify and explain your work.

## Do the paths intersect?

We now want to find out if the shark and Susie are at the same point *at the same time*, which is different from the two paths intersecting. We experience this difference every day we drive; when we come to an intersection, our path (the road we are on) intersects the paths of other cars (on a road approximately perpendicular to our path). We hope, however, that we are not at the same point of the intersection at the same point in time.

To check this, we need to find a time, *t*, such that **x**(t)=**z**(t), i.e. (x(t), y(t), z(t)) = (x2(t), y2(t), z2(t)). To check this, we will solve the three equations, and check for whether the paths share the same coordinates at the same point in time.

```
> solve([x(t) = x2(t), y(t) = y2(t), z(t) = z2(t)], t);
```

$$\{t = 4\pi\} \quad \textbf{(2.2.1)}$$

We now check our answer.

```
> LocSusie := <x(4*Pi), y(4*Pi), z(4*Pi)>;
  LocShark := <x2(4*Pi), y2(4*Pi), z2(4*Pi)>;
```

$$LocSusie := \begin{bmatrix} 0 \\ 0 \\ 2 \end{bmatrix}$$

$$LocShark := \begin{bmatrix} 0 \\ 0 \\ 2 \end{bmatrix} \quad \textbf{(2.2.2)}$$

So we see that at t = 4π the shark and Susie coincide... Sounds bad for the fish.

### Exercises

**#8)** Now that we know the shark and Susie meet, will Susie be swimming toward the shark or away from the shark at that point in time? Be sure to explain all your work.

# Lizzie the Guppy

Susie's friend Lizzie is in a hurry to get to school. She doesn't have time to swim in lazy loops like Susie. Her path of travel is given by **y**(t)=($t^3$-6$t^2$+12t-8, $t^2$-4t+4,$t^3$-6$t^2$+7t+4). She starts out at t=0. Her school is located at the position (27, 9, 14). Repeat the process you followed in studying Susie's path to study the path traveled by Lizzie:

## Exercises

**#9)** Plot the projection of Lizzie's path on each of the coordinate planes. Provide an animation of these paths, and use it to find Lizzie's starting location and direction of travel.

**#10)** Based on the plots from #1, do your best to describe Lizzie's path of travel.

**#11)** Use the spacecurve command to plot Lizzie's path in full 3d. Describe her path of travel, and compare the result to your conclusions in #2.

**#12)** Consider the same Leisure Shark that may have eaten Susie. Does he eat Lizzie? Why or why not? Justify your response with both a plot (animation may help) and an explanation in full sentences.

## There's Something in the Water: Extra Credit

Tommy the Baracuda hadn't felt well all day. His heart was racing and he felt dizzy. He finally went to the school nurse who told him, "there must be something in the water.
Everyone seems to be sick with this strange illness." And with that, she sent him home. His path of travel is given by **r**(t) = ( (4+sin(20t))cos(t), (4+sin(20t))sin(t), cos(20t)) for
$0 \leq t \leq 2\pi$.

### Exercises: Tommy the Baracuda

#13) Plot the projections of Tommy's path on the x/y, y/z, and x/z planes. Animate these paths. Can you guess the 3D motion from these projections?

>

#14) Plot and animate the spacecurve and compare it to your guess in exercise 1.

>

#15) A small, delicious fish is swimming in the same vicinity as Tommy. His path of travel is given by **r**(t) = (5sint, cost, 1) for $0 \leq t \leq 2\pi$. Please plot and animate this curve on the same graph as your spacecurve from exercise 2. Will Tommy eat this fish on his whirling attempt to get home? Prove your answer algebraically.

>